
Örnek problem 
 

Bir çalışmada öğrenciler ile ilgili bilgiler OGRENCI adlı rasgele bir kütükte saklanmak ve daha 

sonra da işlenmek isteniyor. Kütüğün tutanak bilgileri aşağıda yer almaktadır. 

 

 Öğrenci numarası 13 karakter alfa sayısal 

 Öğrenci adı  10 karakter alfa sayısal 

 Öğrenci soy adı 10 karakter alfa sayısal 
 1. Vize notu     3 karakter sayısal 

 2. Vize notu    3 karakter sayısal 

 Final notu    3 karakter sayısal 

 Silindi-kodu    1 karakter sayısal 
  

(Silindi kodu, kaydı silinen öğrenciler için 2, faal olanlar için 1 dir) 

Aşağıda listelenen işleri gerçekleştiren QBasic programlarını kodlayınız. 

 
 (1) OGRENCI  kütüğünü yaratan ve kütüğe tutanak ekleyen. 

  (2) OGRENCI kütüğünden tutanak silen. 

 (3) OGRENCI kütüğünü numaraya göre sıraya dizen. 

 (4) OGRENCI kütüğünü soyadına göre sıraya dizen. 

 (5) OGRENCI kütüğünü listeleyen. 
 

 

Problem ile ilgili programlarda, değişken adları, tampon bellek ve program değişkenleri olarak 

aşağıda verildiği biçimde kullanılacaktır. 

 

Değişken adı    Programdaki biçimi     Tampon bellekteki biçimi 
------------------------------------------------------------------------------ 

Öğrenci no PRGOGNO$ TAMOGNO$ 

Öğrenci adı PRGADI$ TAMADI$ 

Öğrenci soy adı PRGSOYAD$ TAMSOYAD$ 

1. Vize notu PRGVIZE1 TAMVIZE1 

2. Vize notu PRGVIZE2 TAMVIZE2 

Final notu PRGFIN TAMFIN 

Silindi-kodu PRGSIL TAMSIL 

------------------------------------------------------------------------------ 
 

Çözüm-1 
'------------------------------------------------------- 

'   Bu program rasgele OGRENCI kütüğünü yaratır 

'------------------------------------------------------- 

        TYPE OGRTUTANAK 

                OGNO  AS STRING * 13 

                ADI   AS STRING * 10 

                SOYAD AS STRING * 10 

                VIZE1 AS INTEGER 

                VIZE2 AS INTEGER 

                FIN   AS INTEGER 

                SIL   AS INTEGER 

        END TYPE 

   DIM OGRENCI AS OGRTUTANAK 

   OPEN "OGRENCI" FOR RANDOM AS #1 LEN = LEN(OGRENCI) 

   OPEN "R", #2, "TUTSAY" 

   FIELD #2, 6 AS TUT$ 

   GET #2, 1 

        TSAY% = CVI(TUT$) 

        KONTROL = 1 

        WHILE KONTROL = 1 

            GOSUB ACIKLAMA.YAZ 

            GOSUB BILGI.GIR 



            GOSUB VAR.KONTROL 

            IF VAR = 0 THEN GOSUB TUT.YAZ 

            GOSUB DEVAM.SOR 

        WEND 

        GOSUB KUTUK.KAPAT 

        END 

TUT.YAZ: 

'---- TUTANAK SAYISI KÜTÜĞÜNE YAZILIYOR ---------------- 

        TSAY% = TSAY% + 1 

        LSET TUT$ = MKI$(TSAY%) 

        PUT #2, 1 

'---- VERİ KÜTÜĞÜNE YAZILIYOR -------------------------- 

        GOSUB TAMPONA.AL 

        PUT #1, TSAY%, OGRENCI 

RETURN 

ACIKLAMA.YAZ: 

  CLS 

  LOCATE 8, 10: PRINT "ÖĞRENCİ BİLGİLERİ GİRİŞİ" 

  LOCATE 9, 10 

  PRINT "----------------------------------------" 

  LOCATE 10, 10: PRINT "ÖĞRENCİ NUMARASI.............>" 

  LOCATE 11, 10: PRINT "AD...........................>" 

  LOCATE 12, 10: PRINT "SOYAD .......................>" 

  LOCATE 13, 10: PRINT "1. VİZE NOTU.................>" 

  LOCATE 14, 10: PRINT "2. VİZE NOTU.................>" 

  LOCATE 15, 10: PRINT "FİNAL NOTU...................>" 

  LOCATE 10, 50: PRINT "[..............]" 

  LOCATE 11, 50: PRINT "[...........]" 

  LOCATE 12, 50: PRINT "[...........]" 

  LOCATE 13, 50: PRINT "[...]" 

  LOCATE 14, 50: PRINT "[...]" 

  LOCATE 15, 50: PRINT "[...]" 

RETURN 

BILGI.GIR: 

     LOCATE 10, 51: INPUT "", PRGOGNO$ 

     LOCATE 11, 51: INPUT "", PRGADI$ 

     LOCATE 12, 51: INPUT "", PRGSOYAD$ 

     LOCATE 13, 51: INPUT "", PRGVIZE1 

     LOCATE 14, 51: INPUT "", PRGVIZE2 

     LOCATE 15, 51: INPUT "", PRGFIN 

     PRGSIL = 1 

RETURN 

DEVAM.SOR: 

     CEVAP = 1 

     WHILE CEVAP = 1 

     LOCATE 20, 10 

     INPUT "Başka öğrenci girişecek mi?  (E/H)....>", C$ 

              IF C$ = "E" OR C$ = "e" THEN 

                      KONTROL = 1 

                      CEVAP = 0 

              ELSE 

                   IF C$ = "H" OR C$ = "e" THEN 

                       KONTROL = 0 

                       CEVAP = 0 

                   END IF 

              END IF 

        WEND 

RETURN 

TAMPONA.AL:                       ‘tutanağa alınıyor 

         OGRENCI.OGNO  = PRGOGNO$ 

         OGRENCI.ADI   = PRGADI$ 



         OGRENCI.SOYAD = PRGSOYAD$ 

         OGRENCI.VIZE1 = PRGVIZE1 

         OGRENCI.VIZE2 = PRGVIZE2 

         OGRENCI.FIN   = PRGFIN 

         OGRENCI.SIL   = PRGSIL 

RETURN 

VAR.KONTROL: 

   VAR = 0 

   FOR I% = 1 TO TSAY% 

      GET #1, I%, OGRENCI 

   IF PRGOGNO$ = RTRIM$(OGRENCI.OGNO) THEN 

      LOCATE 19, 10 

      PRINT "Bu öğrencinin kaydı var. Tekrar girilemez" 

      LOCATE 20, 10 

      PRINT "Devam etmek için herhangi bir tuşa basınız" 

      LOCATE 20, 75 

      LOCATE 20, 76 : INPUT "", DEVAM$ 

      VAR = 1 

      LOCATE 19, 1: PRINT SPACE$(80) 

      LOCATE 20, 1: PRINT SPACE$(80) 

      EXIT FOR 

   ELSE 

   END IF 

   NEXT I% 

RETURN 

KUTUK.KAPAT: 

        CLOSE #1 , #2 

RETURN 

 


